

```

1  'WALLS 250519
10 CLV:CLP:CLS
20 LET[0],30,30,14,14,21,21,14,14:C=6:P=12:Y=15:J=226
30 W="=====
40 POKE#700,4,76,68,228,68,68,14,0,4,74,66,226,68,72,14,0,26,26,60,88,152,3
6,36,4,88,88,60,26,25,36,36,32
50 GSB200:LCX,Y:?"▲":LC13,9:?"READY!":R=SCR(RND(6)+13,9)
60 PLAY"T36004EGGRE+AARB+GAB<C":WAIT140:LCX,Y:?"·":CLT
100 GSB200:C=(C+1)%P:M=(M+1)%8
110 IF!CCOPY#900,W+RND(28),32:P=8+RND(6)
120 IF!MLCRND(30)+1,0:?"♥"
130 SCROLL2:Q=SCR(X,Y):BEEP[M]
140 IFQ=6GOTO400
150 IFQ=32BEEP:S=S+1:J=224
160 IFQ=229ORQ=RBEEP7:S=S+2:J=225
170 B=900-TICK()/6:GSB300:IFB=206GOTO500
180 FORA=0TOB:NEXT:IFJ=226+(M+1)%2:LCX,Y:?"·"
190 J=226+M%2:GOTO100
200 X=(ANA(2)-16)/32:IFX<1X=1ELSEIFX>30X=30
210 RTN
300 LCX,Y:?"CHR$(J):LC2,23:?"S":LC28,23:?"(B-200)*10/70":RTN
400 J=226:GSB300:BEEP40,30:WAIT30:J=250:GSB300:BEEP24:WAIT6:BEEP6:LC12,9:?"
GAME OVER":CLK
410 IFINKEY()=10ORBTN()RUNELSECONT
500 PLAY"T30004GG-G<C2.>G2.E2.E+FG<C1":WAIT250:J=253:S=S+50:GSB300:LC8,9:?"
CONGRATULATIONS!":CLK:GOTO410

```

A : 汎用 (FOR~NEXT ループなど)
B : スピード調整、ゴールまでの残り時間算出用
C : 壁出現カウンター
J : プレイヤーとして表示するキャラクターのコード
M : BGM、♥ 出現カウンター ([M] : BGM の音程)
P : 次の壁出現の間隔の値
Q : プレイヤー移動先に表示されているキャラクターのコード
R : 隠しボーナスのキャラクターのコード
S : スコア
X : プレイヤーの X 座標
Y : プレイヤーの Y 座標

10～60 行：初期設定(変数リセット、BGM の音程書き込み、得点のキャラクター定義、スタート画面表示)
100～190 行：メインループ
200～210 行：アナログコントローラー値読み込み
300 行：プレイヤー、スコア、残り時間表示
400～410 行：ゲームオーバー処理
500 行：クリアエンディング表示

IchigoJam S など 1.4 系で動かす場合は 170 行、180 行、300 行を下記のように変更してください。

★ 30 行の壁のキャラクター()は、キーボードから直接出すことができません。

★ IchigoJam 本体と可変抵抗器(ポテンシオメータ)との接続は VCC-VCC、GND-GND、IN2-OUT、(アナログコントローラがあれば)BTN-ショットボタンで繋いでください。(接続端子が一部異なりますが、過去作『WALLS』の PDF ファイルに写真を掲載しております。ご参考になさってください。)